clear all
set more off
set mem 10000000
set matsize 10000
version 15

******************************************************************************************
*** RDROBUST reduced-form, SECC outcomes: including states with bad/missing shapefiles ***
******************************************************************************************

** Set file paths
do "$path_code/paths.do"

** Set graph scheme
cd "$path/code/analyze"
set scheme fb, perm

****************************************************************** 
****************************************************************** 

{
use "$panel/panel_dataset_full.dta", clear
drop hpca11* vd*11

	// Keep villages in RD sample (INCLUDING THE 10 STATES WITH BAD/MISSING SHAPEFILES
gen in_rf_sample = vplan4<11 & sample_h==1 & sing_h==1 & pop_non_zero==1 
keep if in_rf_sample==1 

	// Merge in SECC outcomes at the village level
merge 1:m pca01_id using "$panel/secc_pca_vill_all.dta", keep(3) nogen
gen p_ratio = n_tot/tot_p11
gen hh_ratio = secc_n_hh/no_hh11
keep if p_ratio<=1.10
keep if hh_ratio<=1.10
	
	// Create state FEs (since RD robust doesn't let you pass them through)
drop if st_code==32 // Kerala, only 3 villages
drop if st_code==11 // Sikkim, only 5 villages
tab st_code if in_rf_sample==1, gen(STFErf)	
drop STFErf1 STFErf2 STFErf4 // to avoid collinearity (also drop Haryana and Himachal Pradesh FEs, since there are too few observations)

	// Create district FEs (since RD robust doesn't let you pass them through)
gen stdtFE = stdt
tab stdtFE state     if inlist(stdtFE,15,62,63,64,70,91,101,116,117,138,158,175,180,181,202,209,213,288,308,309,320,433,490,493,494,495,504,505,508)	
replace stdtFE = 999 if inlist(stdtFE,15,62,63,64,70,91,101,116,117,138,158,175,180,181,202,209,213,288,308,309,320,433,490,493,494,495,504,505,508)	
	// one catch-all district FE for districts with so few in-sample villages that they break rdrobust
tab stdtFE if in_rf_sample==1, gen(DTFErf)	
drop DTFErf1 // to avoid collinearity

	// Create block groups (for clustering)
egen stdtbk = group(stdt bk_code)

// 	// Create lights-difference variable, to identify crazy outliers
// gen lights_diff = abs(lights_max2011_hat - lights_max2001_hat)

	// Create apples-to-apples "other" category
foreach g in m f {
	foreach a in _ _16_ {
		gen n_`g'`a'occ_ot = n_`g'`a'occ_oth + n_`g'`a'occ_wrk
	}
}	

	// Definte SECC labor shares for whole population
foreach g in m f {
	gen denom_`g' = n_`g'_age_0_6 + n_`g'_age_7_15 + n_`g'_age_16_25 + n_`g'_age_26_40 + n_`g'_age_41_60 + n_`g'_age_61_plus
	gen work_secc_ag_`g' = n_`g'_occ_agr/denom_`g'
	gen work_secc_hh_`g' = n_`g'_occ_dom/denom_`g'
	gen work_secc_ot_`g' = n_`g'_occ_ot/denom_`g'
	gen work_secc_wk_`g' = n_`g'_occ_wrk/denom_`g'
	gen work_secc_oth_`g' = n_`g'_occ_oth/denom_`g'
}	
	
  // Define SECC-specific reduced-form outcome variables
foreach g in m f {
	foreach v of varlist n_`g'_16_occ* {
		local v1 = subinstr("`v'","_16_","_",1)
		local v2 = subinstr("`v'","_16_","_youth_",1)
		gen `v2' =  `v1' - `v'
		assert `v'!=. & `v2'!=.
		replace `v' = `v'/(n_`g'_age_16_25 + n_`g'_age_26_40 + n_`g'_age_41_60 + n_`g'_age_61_plus)
		replace `v2' = `v2'/(n_`g'_age_0_6 + n_`g'_age_7_15) 
		replace `v' = 0 if `v'==.
		replace `v2' = 0 if `v2'==.
		assert `v'<=1 & `v2'<=1 & `v'>=0 & `v2'>=0 
	}
	foreach v of varlist n_`g'_16_single n_`g'_16_illit n_`g'_16_midsch {
		replace `v' = `v'/(n_`g'_age_16_25 + n_`g'_age_26_40 + n_`g'_age_41_60 + n_`g'_age_61_plus)
		replace `v' = 0 if `v'==.
		assert `v'<=1 & `v'>=0 
	}	
	
	gen n_`g'_16_emp = n_`g'_16_occ_agr + n_`g'_16_occ_dom + n_`g'_16_occ_oth + n_`g'_16_occ_wrk
	gen n_`g'_youth_emp = n_`g'_youth_occ_agr + n_`g'_youth_occ_dom + n_`g'_youth_occ_oth + n_`g'_youth_occ_wrk	
}	

	// Main source of HH income: combine foraging and "other" into "other" category
replace pct_hh_mnth_inc_oth = pct_hh_mnth_inc_oth + pct_hh_mnth_inc_for
replace wpct_hh_mnth_inc_oth = wpct_hh_mnth_inc_oth + wpct_hh_mnth_inc_for
	
	// Convert illiterary to literacy
foreach v of varlist *illit {
	local v2 = subinstr("`v'","illit","lit",1)
	gen `v2' = 1 - `v'
	drop `v'
}	

	// Household head apples-to-apples other category
gen pct_hhh_occ_ot = pct_hhh_occ_oth + pct_hhh_occ_wrk

	// Drop outcomes that aren't useful and will just make the regression loop longer
drop *pct_hh_mnth_inc_10 *pct_hh_mnth_inc_for

	// Create macros for lists of outcome variables 
global yvars = "hh_ratio pct_hh_mnth_inc_5_10 pct_hh_salaried_job pct_hh_own_any_land n_m_16_occ_agr n_f_16_occ_agr n_m_16_occ_dom n_f_16_occ_dom n_m_16_occ_ot n_f_16_occ_ot "

	// Merge in SHRUG outcomes to create index variable
merge 1:1 pca01_id using "$shrug/shrug_secc.dta", keep(1 3) nogen
gen hh_ratio_conv = max(1-hh_ratio,0)
gen secc11_inc_cultiv_share_conv = max(1-secc11_inc_cultiv_share,0)
foreach v of varlist hh_ratio_conv pct_hh_mnth_inc_5_10 pct_hh_salaried_job pct_hh_own_any_land secc11_inc_cultiv_share_conv {
	qui sum `v' if tot_p<=1000
	gen Z`v' = (`v' - r(mean))/(r(sd))
}
egen double index_secc11 = rmean(Z*)
drop Z*


	// Create variables to store regresson results
gen rf_step = .
gen yvar = ""
gen ifs = ""
gen controls_base = ""
gen control = ""
gen fe = ""
gen kernel = ""
gen bwmethod = ""
gen vce = ""
gen polynomial_order = .
gen beta_conv = .
gen beta_robust = .
gen se_conv = .
gen se_robust = .
gen pval_conv = .
gen pval_robust = .
gen lci_conv = .
gen uci_conv = .
gen lci_robust = .
gen uci_robust = .
gen bw_lo = .
gen bw_hi = .
gen nobs_orig = .
gen nobs_left = .
gen nobs_right = .
gen nobs_total = .
gen ndist = . 
gen nstate = . 
gen ymean = .
gen ftag = ""
global h_wide = 200


	// Massive loop over RDROBUST outcomes and sensitivities
foreach y in $yvars index_secc11 {

	// Prep to store results
	foreach v of varlist rf_step-ftag {
		cap replace `v' = ""
		cap replace `v' = .
	}
	local row = 0

	// Define outcome-specific stuff
	global yvar = "`y'"
	global control = subinstr("$yvar","11","01",1)
	if substr("$yvar",-4,4)=="1101" {
		global control = subinstr("$yvar","1101","01",1)
	}
	do "$path/code/analyze/RDROBUST_rf_outcomes_graphspecs.do"

	//Loop through sensitivities
	foreach rf_step in 1 4 5 11 {

		// Reset RDROBUST defaults
		local ifs = "in_rf_sample==1 & pop_mismatch20==0"
		local controls_base = "" 
		local fe = "STFErf*"
		local kernel = "tri"
		local bwmethod = "mserd"
		local vce = ""
		local poly = 1
		local graphtag = "rf"

		// Define step-specific RDROBUST settings
		if inlist(`rf_step',1) {
			local folder = "RDROBUST plots rf outcomes preferred"
			local ftag = "RF outcomes, preferred, including bad shapfiles"
			local title2 = ""
		}
		if inlist(`rf_step',2) {
			local folder = "RDROBUST plots rf outcomes popmismatch"
			local ftag = "RF outcomes, popmismatch, including bad shapfiles"
			local title2 = "(population mismatches)"
			local ifs = "in_rf_sample==1"
		}
		if inlist(`rf_step',4) {
			local folder = "RDROBUST plots rf outcomes epa kernel"
			local ftag = "RF outcomes, epa kernel, including bad shapfiles"
			local title2 = "(Epanechnikov kernel)"
			local kernel = "epa"
		}
		if inlist(`rf_step',5) {
			local folder = "RDROBUST plots rf outcomes uni kernel"
			local ftag = "RF outcomes, uni kernel, including bad shapfiles"
			local title2 = "(uniform kernel)"
			local kernel = "uni"
		}
		if inlist(`rf_step',7) {
			local folder = "RDROBUST plots rf outcomes no FEs"
			local ftag = "RF outcomes, no FEs, including bad shapfiles"
			local title2 = "(no fixed effects)"
			local fe = ""
		}
		if inlist(`rf_step',8) {
			local folder = "RDROBUST plots rf outcomes district FEs"
			local ftag = "RF outcomes, district FEs, including bad shapfiles"
			local title2 = "(district fixed effects)"
			local fe = "DTFEfs*"
		}
		if inlist(`rf_step',9) {
			local folder = "RDROBUST plots rf outcomes nncluster by district"
			local ftag = "RF outcomes, nncluster by district, including bad shapfiles"
			local title2 = "(nncluster by district)"
			local vce = "vce(nncluster stdt)"
		}
		if inlist(`rf_step',10) {
			local folder = "RDROBUST plots rf outcomes cluster by district"
			local ftag = "RF outcomes, cluster by district, including bad shapfiles"
			local title2 = "(cluster by district)"
			local vce = "vce(cluster stdt)"
		}
		if inlist(`rf_step',11) {
			local folder = "RDROBUST plots rf outcomes CERRD bandwidth"
			local ftag = "RF outcomes, CERRD bandwidth, including bad shapfiles"
			local title2 = "(CERRD bandwidth)"
			local bwmethod = "cerrd"
		}

		// Run first-stage regresssion
		di "rdrobust $yvar tot_p if `ifs', c(299.5) covs(`controls_base' $control `fe') kernel(`kernel') bwselect(`bwmethod') p(`poly') all `vce'"
		rdrobust $yvar tot_p if `ifs', c(299.5) covs(`controls_base' $control `fe') kernel(`kernel') bwselect(`bwmethod') p(`poly') all `vce'

		// Generate in-sample indicator and store bandwidths
		cap drop temp_in_reg
		qui gen temp_in_reg = `ifs' & inrange(tot_p,299.5-e(h_l),299.5+e(h_r)) & ${yvar}!=.
		local h_l = e(h_l)
		local h_r = e(h_r)

		// Store results
		local row = `row' + 1
		qui replace rf_step = `rf_step' in `row'
		qui replace yvar = "$yvar" in `row'
		qui replace ifs = "`ifs'" in `row'
		qui replace controls_base = "`controls_base'" in `row'
		qui replace control = "$control" in `row'
		qui replace fe = "`fe'" in `row'
		qui replace kernel = e(kernel) in `row'
		qui replace bwmethod = e(bwselect) in `row'
		qui replace vce = "`vce'" in `row'
		qui replace polynomial_order = e(p) in `row'
		qui replace beta_conv = e(tau_cl) in `row'
		qui replace beta_robust = e(tau_bc) in `row'
		qui replace se_conv = e(se_tau_cl) in `row'
		qui replace se_robust = e(se_tau_rb) in `row'
		qui replace pval_conv = e(pv_cl) in `row'
		qui replace pval_robust = e(pv_rb) in `row'
		qui replace lci_conv = e(ci_l_cl) in `row'
		qui replace uci_conv = e(ci_r_cl) in `row'
		qui replace lci_robust = e(ci_l_rb) in `row'
		qui replace uci_robust = e(ci_r_rb) in `row'
		qui replace bw_lo = e(h_l) in `row'
		qui replace bw_hi = e(h_r) in `row'
		qui replace nobs_orig = e(N) in `row'
		qui replace nobs_left = e(N_h_l) in `row'
		qui replace nobs_right = e(N_h_r) in `row'
		qui replace nobs_total = e(N_h_l) + e(N_h_r) in `row'
		qui unique stdt if temp_in_reg==1
		qui replace ndist = r(unique) in `row'
		qui unique st_code if temp_in_reg==1
		qui replace nstate = r(unique) in `row'
		qui sum $yvar if temp_in_reg==1 & tot_p<299.5
		qui replace ymean = r(mean) in `row'
		qui replace ftag = "`ftag'" in `row'

		// Residualize $yvar for regression sample
		qui reg $yvar `controls_base' $control `fe' if temp_in_reg==1
		cap drop y_resid_sample
		predict y_resid_sample, residuals
		
		// Residualize $yvar for [100,500] bandwidth
		qui reg $yvar `controls_base' $control `fe' if `ifs' & inrange(tot_p,299.5-${h_wide},299.5+${h_wide})
		cap drop y_resid_full
		predict y_resid_full, residuals
		
		// RD plots for regression sample (variable bandwidth) 
		foreach bins in 10 20 40 {
			if `h_l'<=75 {
				local xlab = "250 275 300 325 350"
			}
			else if inrange(`h_l',76,125) {
				local xlab = "200 250 300 350 400"
			}
			else if inrange(`h_l',126,175) {
				local xlab = "150 200 250 300 350 400 450"
			}
			else {
				local xlab = ""
			}
			rdplot y_resid_sample tot_p if temp_in_reg==1, ///
				c(299.5) p(`poly') nbins(`bins' `bins') kernel(`kernel') h(`h_l' `h_r') ///
				graph_options( ///
				title("$title" "`title2'", color(black) size(large)) ///
				ytitle("$ytitle", size(medlarge)) ///
				xtitle("2001 village population", size(medlarge)) ///
				ylabel(,nogrid angle(0) labsize(medlarge)) ///
				xlabel(`xlab', labsize(medlarge)) ///
				graphregion(color(white)) plotregion(fcolor(white)) graphregion(lcolor(white)) ///
				legend(off))
			graph export "$results/`folder'/${yvar}_`graphtag'_inreg_`bins'.pdf", replace
		}

		if !inlist(`rf_step',11) {
			// RD plots for regression sample (constant bandwidth) 
			foreach bins in 20 40 {
				rdplot y_resid_full tot_p if `ifs' & inrange(tot_p,299.5-${h_wide},299.5+${h_wide}), ///
					c(299.5) p(`poly') nbins(`bins' `bins') kernel(`kernel') h(${h_wide} ${h_wide}) ///
					graph_options( ///
					title("$title" "`title2'", color(black) size(large)) ///
					ytitle("$ytitle", size(medlarge)) ///
					xtitle("2001 village population", size(medlarge)) ///
					ylabel(,nogrid angle(0) labsize(medlarge)) ///
					xlabel(, labsize(medlarge)) ///
					graphregion(color(white)) plotregion(fcolor(white)) graphregion(lcolor(white)) ///
					legend(off))
				graph export "$results/`folder'/${yvar}_`graphtag'_wide_`bins'.pdf", replace
			}
		}
	}

	// Save results
	preserve
	keep rf_step-ftag
	dropmiss, obs force
	if "$yvar"!="hh_ratio" {
		tempfile reg_results
		save `reg_results'
		clear
		append using "$results/RDROBUST_outcomes_rf_secc_badshapefiles.dta" `reg_results'
	}
	duplicates drop
	compress
	save "$results/RDROBUST_outcomes_rf_secc_badshapefiles.dta", replace
	restore
	
}



}

****************************************************************** 
****************************************************************** 

